**Name** ____________________________ Class Time _____

**Math 2250, Numerical Methods**
**Maple Project Sample Solution**
**Spring 2012**

**References**: Code in `maple` appears in `2250mapleL4-sample-S2012.txt` at URL `http://www.math.utah.edu/~gustafso/`. This document: `2250mapleL4-sample-S2012.pdf`. Other related and required documents are available at the web site:

- Numerical Solution of First Order DE (typeset, 19 pages, 220k pdf). A resource similar to the textbook, with maple examples and deeper detail. It is for a second reading, in case Edwards-Penney left too many questions unanswered.
- Numerical DE coding hints, `2250numerical-hints.txt`, TEXT Document (1 page, 2k). A modified portion of this document is appended here, for completeness.
- The web copy `2250mapleL4-sample-S2012.txt` of the text in this document is suited for mouse copy and paste of maple code segments.

## Problem ER-2. (E & P Exercise 2.6-36, Symbolic Solution)

The exact symbolic solution of the Logistic problem $y' = 0.02225y - 0.0003y^2$, $y(0) = 25$ is

$$y(x) = \frac{2225}{30 + 59e^{-89x/4000}}.$$

Using textbook techniques, Chapter 2, derive the answer. Then check the answer in maple.

**Solution**.
**Derivation Details**. The differential equation is a Verhulst-Logistic equation, studied in Section 2.1, appearing as equation (6):

$$\frac{dy}{dx} = ky(M-y), \quad kM = 0.02225, \quad k = 0.0003.$$

The unique solution $y(x)$ with $y(0) = y_0$ is given by equation (7):

$$y(x) = \frac{My_0}{y_0 + (M - y_0)e^{-kMx}}.$$

The fraction will be multiplied top and bottom by the factor $k/y_0$, to obtain

$$\begin{aligned}
y(t) &= \frac{k/y_0}{k/y_0}\,\frac{My_0}{y_0 + (M-y_0)e^{-kMx}} \\
&= \frac{kM}{k + (kM/y_0 - k)e^{-kMx}} \\
&= \frac{0.02225}{0.0003 + (0.02225/25 - 0.0003)e^{-0.02225x}} \\
&= \frac{100000}{100000}\,\frac{0.02225}{0.0003 + (0.02225/25 - 0.0003)e^{-0.02225x}} \\
&= \frac{2225}{30 + 59e^{-89x/4000}}.
\end{aligned}$$

**Answer Check in Maple**.

```
# Check the exact symbolic solution
 de:=diff(y(t),t)=0.02225 *y(t) - 0.0003*y(t)^2;
 ic:= y(0)=25;
 dsolve({de,ic},y(t));
```

*Staple this page on top of your hand-written report*

**Name** ____________________________ Class Time _____

**Maple Project Sample Solution: Numerical Methods**
**Math 2250 S2012**

**References**: Code in `maple` appears in `2250mapleL4-sample-S2012.txt` at URL `http://www.math.utah.edu/~gustafso/`. This document: `2250mapleL4-sample-S2012.pdf`. Further references appear on the previous page.

## Problem L4.1. (E & P Exercise 2.6-36)

Consider the initial value problem $y' = 0.02225y - 0.0003y^2$, $y(0) = 25$ with symbolic solution $y(t) = \dfrac{2225}{30 + 59e^{-89t/4000}}$. Apply Euler's method to finds the numerical solution $y(x)$ on $x = 0$ to $x = 250$. Write computer code to produce two dot tables. The first has $n + 1 = 101$ rows, $h = 250/n = 2.5$. The second has $n + 1 = 201$ rows, $h = 250/n = 1.25$. The computation should find the missing digits in the table below.

| $x$ | 0.0 | 50 | 100 | 150 | 200 | 250 |
|---|---|---|---|---|---|---|
| $y$-approx, $h = 2.5$ | 25.00000000 | 45.0101???? | 61.2965???? | 69.4877???? | 72.6063???? | 73.6622???? |
| $y$-approx, $h = 1.25$ | 25.00000000 | 45.0280???? | 61.2316???? | 69.4052???? | 72.5539???? | 73.6367???? |
| actual $y(x)$ | 25.00000000 | 45.04465339 | 61.16674082 | 69.32324992 | 72.50146404 | 73.61087799 |
| Error(approx,actual) | 0.0000% | 0.03??% | 0.10??% | 0.11??% | 0.07??% | 0.03??% |

---

**Part I**. Reproduce, by transcribing computer data, the table above, and fill in missing digits. For the percentage error with $h = 250/200 = 1.25$, use the equation

$$Error(approx, actual) = 100\frac{|approx - actual|}{|actual|}.$$

**Solution**.

$y$-approx, $h = 2.5$, 25.00000000, 45.01012660, 61.29651142, 69.48777402, 72.60632272, 73.66229582.

$y$-approx, $h = 1.25$, 25.00000000, 45.02802159, 61.23165186, 69.40522495, 72.55394452, 73.63678526.

Symbolic $y(x)$, 25.00000000 45.04465339 61.16674082 69.32324992 72.50146404 73.61087799.

Error(approx,actual), $h = 1.25$, percentages 0.0, 0.03692291704, 0.1061214626, 0.1182504140, 0.07238540724, 0.03519489335.

---

**Part II**. Hand-check the first dot table for one step. The answer should be the same as line 2 of the first dot table (which has 101 lines). Assume the given symbolic solution is correct. Don't repeat details already done in ER-2. Test the answers against the symbolic solution, as suggested in the table above.

**Hand Check for Euler**.

```
One step.
h=2.5
x0 = 0
y0 = 25
f(x,y) = 0.02225 y - 0.0003 y^2
y1 = y0 + h f(x0,y0)
   = 25 + 2.5 (0.02225 (25) - 0.0003 (25)^2)
   = 25.921875
Dots[1] = [0, 25], Dots[2] = [2.500000000, 25.92187500]. Answer checks.
```

**Symbolic Solution Check**.

The Euler answer and the symbolic answer **agree to one digit**.

---

**Part III**. Include an appendix of the computer code used.

```
# Now for the Euler code to make the dot table, error percentages and plot.
  # Euler. Group 1, initialize.
   f:=(x,y)->0.02225 *y - 0.0003*y^2;
   x0:=0:y0:=25:Dots:=[x0,y0]:n:=100:h:=250/n:
  # Group 2, repeat n times. Euler's method
```

```
  for i from 1 to n do
   Y:=y0+h*f(x0,y0);
   x0:=x0+h:y0:=Y:Dots:=Dots,[evalf(x0),evalf(y0)];
  od:
  # Group 3, display relevant dots and plot.
   Exact:=x->2225/(30+59*exp(-89 *x/4000));
   P:=unapply(evalf(100*abs(exact-approx)/abs(exact)),(exact,approx)):
   m:=n/5:X:=[seq(1+m*j,j=0..n/m)]: # List of relevant indices
   print("Dots"),seq(Dots[k],k=X);
   print("Exact"),seq(Exact(Dots[k][1]),k=X);
   print("Error"),seq(P(Exact(Dots[k][1]),Dots[k][2]),k=X);
   #plot([Dots]);
### The output from this program:
                              "Dots"
  [0, 25], [50., 45.01012660], [100., 61.29651142], [150., 69.48777402],
  [200., 72.60632272], [250., 73.66229582]
                             "Exact"
  25, 45.04465339, 61.16674082, 69.32324992, 72.50146404, 73.61087799
                             "Error"
  0., 0.07665014025, 0.2121587619, 0.2373288907, 0.1446297415, 0.06985085819
```

## Problem L4.2. (E & P Exercise 2.6-36)

Consider the initial value problem $y' = 0.02225y - 0.0003y^2$, $y(0) = 25$ with symbolic solution $y(t) = \dfrac{2225}{30 + 59e^{-89t/4000}}$. Apply Heun's method to finds the numerical solution $y(x)$ on $x = 0$ to $x = 250$. Write computer code to produce two dot tables. The first has $n + 1 = 101$ rows, $h = 250/n = 2.5$. The second has $n + 1 = 201$ rows, $h = 250/n = 1.25$. The computation should find the missing digits in the table below.

| $x$ | 0.0 | 50 | 100 | 150 | 200 | 250 |
|---|---|---|---|---|---|---|
| $y$-approx, $h = 2.5$ | 25.00000000 | 45.0419???? | 61.1624???? | 69.3195???? | 72.4992???? | 73.6098????. |
| $y$-approx, $h = 1.25$ | 25.00000000 | 45.0439???? | 61.1656???? | 69.3223???? | 72.5009???? | 73.6106????. |
| actual $y(x)$ | 25.00000000 | 45.04465339 | 61.16674082 | 69.32324992 | 72.50146404 | 73.61087799 |
| Error(approx,actual) | 0.0000% | 0.001??% | 0.001??% | 0.001??% | 0.000??% | 0.000??% |

**Part I**. Reproduce, by transcribing computer data, the table above, and fill in missing digits. For the percentage error with $h = 250/200 = 1.25$, use the equation

$$Error(approx, actual) = 100\frac{|approx - actual|}{|actual|}.$$

**Solution**.

$y$-approx, $h = 2.5$, 25, 45.04191584, 61.16246299, 69.31954666, 72.49927181, 73.60981811.

$y$-approx, $h = 1.25$, 25, 45.04396719, 61.16567946, 69.32233642, 72.50092484, 73.61061773.

Symbolic $y(x)$, 25, 45.04465339, 61.16674082, 69.32324992, 72.50146404, 73.61087799.

Error(approx,actual), $h = 1.25$, percentages 0.0, 0.001523377245, 0.001735191357, 0.001317739721, 0.0007437091197, 0.0003535618744.

**Part II**. Hand-check the first dot table for one step. The answer should be the same as line 2 of the first dot table (which has 101 lines). Assume the given symbolic solution is correct. Don't repeat details already done in ER-2. Test the answers against the symbolic solution, as suggested in the table above.

**Hand Check for Heun**.

```
One step.
h=2.5
x0 = 0
y0 = 25
f(x,y) = 0.02225 y - 0.0003 y^2
y1 = y0 + h f(x0,y0)
   = 25 + 2.5 (0.02225 (25) - 0.0003 (25)^2)
   = 25.921875
```

```
y2 = y0 + h(f(x0,y0)+f(x0+h,y1))/2
   = 25 + 2.5 (0.02225 (25) - 0.0003 (25)^2)/2 +
     2.5 (0.02225 (25.921875) - 0.0003 (25.921875)^2)/2
   = 25.92991080
Dots[1] = [0, 25], Dots[2] = [2.500000000, 25.92991080]. Answer checks.
```

**Symbolic Solution Check.**

The Heun answer and the symbolic answer **agree to two digits**.

---

**Part III**. Include an appendix of the computer code used.

```
# Now for the Heun code to make the dot table, error percentages and plot.
  # Heun. Group 1, initialize.
   f:=(x,y)->0.02225 *y - 0.0003*y^2;
   x0:=0:y0:=25:Dots:=[x0,y0]:n:=100:h:=250/n:
  # Group 2, repeat n times. Heun's method
  for i from 1 to n do
   Y1:=y0+h*f(x0,y0);
   Y:=y0+h*(f(x0,y0)+f(x0+h,Y1))/2;
   x0:=x0+h:y0:=Y:Dots:=Dots,[evalf(x0),evalf(y0)];
  od:
  # Group 3, display relevant dots and plot.
   Exact:=x->2225/(30+59*exp(-89 *x/4000));
   P:=unapply(evalf(100*abs(exact-approx)/abs(exact)),(exact,approx)):
   m:=n/5:X:=[seq(1+m*j,j=0..n/m)]: # List of relevant indices
   print("Dots"),seq(Dots[k],k=X);
   print("Exact"),seq(Exact(Dots[k][1]),k=X);
   print("Error"),seq(P(Exact(Dots[k][1]),Dots[k][2]),k=X);
   #plot([Dots]);
### The output from this program:
                                "Dots"
  [0, 25], [50., 45.04191584], [100., 61.16246299],
        [150., 69.31954666], [200., 72.49927181], [250., 73.60981811]
                                "Exact"
 25, 45.04465339, 61.16674082, 69.32324992, 72.50146404, 73.61087799
                                "Error"
 0., 0.006077413842, 0.006993719042, 0.005342017295, 0.003023704458,
       0.001439841541
```

## Problem L4.3. (E & P Exercise 2.6-36)

Consider the initial value problem $y' = 0.02225y - 0.0003y^2$, $y(0) = 25$ with symbolic solution $y(t) = \dfrac{2225}{30 + 59e^{-89t/4000}}$. Apply the RK4 method to finds the numerical solution $y(x)$ on $x = 0$ to $x = 250$. Write computer code to produce two dot tables. The first has $n + 1 = 101$ rows, $h = 250/n = 2.5$. The second has $n + 1 = 201$ rows, $h = 250/n = 1.25$. The computation should find the missing digits in the table below.

| $x$ | 0.0 | 50 | 100 | 150 | 200 | 250 |
|---|---|---|---|---|---|---|
| $y$-approx, $h = 2.5$ | 25.00000000 | 45.04465??? | 61.16674??? | 69.32324??? | 72.50146??? | 73.61087???. |
| $y$-approx, $h = 1.25$ | 25.00000000 | 45.04465??? | 61.16674??? | 69.32324??? | 72.50146??? | 73.61087???. |
| actual $y(x)$ | 25.00000000 | 45.04465339 | 61.16674082 | 69.32324992 | 72.50146404 | 73.61087799 |
| Error(approx,actual) | 0.000000% | 0.00000??% | 0.00000??% | 0.00000??% | 0.00000??% | 0.00000??% |

---

**Part I**. Reproduce, by transcribing computer data, the table above, and fill in missing digits. For the percentage error with $h = 250/200 = 1.25$, use the equation

$$Error(approx, actual) = 100\frac{|approx - actual|}{|actual|}.$$

**Solution.**

$y$-approx, $h = 2.5$, 25.0, 45.04465322, 61.16674048, 69.32324952, 72.50146380, 73.61087789.

$y$-approx, $h = 1.25$, 25.0, 45.04465348, 61.16674086, 69.32324992, 72.50146405, 73.61087799.

Symbolic $y(x)$, 25, 45.04465339, 61.16674082, 69.32324992, 72.50146404, 73.61087799.

Error(approx,actual), $h = 1.25$, percentages 0.0, 0.3774032814e-6, 0.5558576368e-6, 0.5770069933e-6, 0.3310277981e-6, 0.1358494868e-6.

---

**Part II**. Assume the given symbolic solution is correct. Don't repeat details already done in ER-2. Test the answers against the symbolic solution, as suggested in the table above.

**Symbolic Solution Check**.

The RK4 answer and the symbolic answer **agree to six digits**.

---

**Part III**. Include an appendix of the computer code used.

```
# Now for the RK4 code to make the dot table, error percentages and plot.
  # RK4. Group 1, initialize.
   f:=(x,y)->0.02225 *y - 0.0003*y^2;
   x0:=0:y0:=25:Dots:=[x0,y0]:n:=100:h:=250/n:
  # Group 2, repeat n times. RK4 method.
 for i from 1 to n do
  k1:=h*f(x0,y0):
  k2:=h*f(x0+h/2,y0+k1/2):
  k3:=h*f(x0+h/2,y0+k2/2):
  k4:=h*f(x0+h,y0+k3):
  Y:=y0+(k1+2*k2+2*k3+k4)/6:
  x0:=x0+h:y0:=Y:Dots:=Dots,[evalf(x0),evalf(y0)];
 od:
  # Group 3, display relevant dots and plot.
   Exact:=x->2225/(30+59*exp(-89 *x/4000));
   P:=unapply(evalf(100*abs(exact-approx)/abs(exact)),(exact,approx)):
   m:=n/5:X:=[seq(1+m*j,j=0..n/m)]: # List of relevant indices
   print("Dots"),seq(Dots[k],k=X);
   print("Exact"),seq(Exact(Dots[k][1]),k=X);
   print("Error"),seq(P(Exact(Dots[k][1]),Dots[k][2]),k=X);
   plot([Dots]);
### The output from this program:
                                "Dots"
  [0, 25], [50., 45.04465322], [100., 61.16674048],
        [150., 69.32324952], [200., 72.50146380], [250., 73.61087789]
                               "Exact"
 25, 45.04465339, 61.16674082, 69.32324992, 72.50146404, 73.61087799
                               "Error"
0.0, .3774032814e-6, .5558576368e-6, .5770069933e-6, .3310277981e-6, .1358494868e-6
```




*Staple this page on top of your hand-written and maple worksheet report*

**End of Sample Maple Lab: Numerical Methods.**

```
# Warning: These snips of code made for y'=1-x-y, y(0)=3.
#          Code computes approx values for y(0.1) to y(1).
# 'Dots' is the list of dots for connect-the-dots graphics.
# ========================================
# Euler. Group 1, initialize.
 f:=(x,y)->1-x-y:
 x0:=0:y0:=3:h:=0.1:Dots:=[x0,y0]:n:=200:
# Group 2, repeat n times. Euler's method
for i from 1 to n do
 Y:=y0+h*f(x0,y0);
 x0:=x0+h:y0:=Y:Dots:=Dots,[x0,y0];
od:
# Group 3, display relevant dots and plot.
 Exact:=x->2-x+exp(-x);
 P:=unapply(evalf(100*abs(exact-approx)/abs(exact)),(exact,approx)):
 m:=40:X:=[seq(1+m*j,j=0..n/m)]: # List of relevant indices
 print("Dots"),seq(Dots[k],k=X);
 print("Exact"),seq(Exact(Dots[k][1]),k=X);
 print("Error"),seq(P(Exact(Dots[k][1]),Dots[k][2]),k=X);
 plot([Dots]);
# ========================================
# Heun. Group 1, initialize.
 f:=(x,y)->1-x-y:
 x0:=0:y0:=3:h:=0.1:Dots:=[x0,y0]:n:=200:
# Group 2, repeat n times. Heun method.
for i from 1 to n do
 Y1:=y0+h*f(x0,y0);
 Y:=y0+h*(f(x0,y0)+f(x0+h,Y1))/2:
 x0:=x0+h:y0:=Y:Dots:=Dots,[x0,y0];
od:
# Group 3, display relevant dots and plot.
 Dots[1],Dots[2],seq(Dots[1+40*j],j=1..n/40);
 plot([Dots]);

# ========================================
# RK4. Group 1, initialize.
 f:=(x,y)->1-x-y:
 x0:=0:y0:=3:h:=0.1:Dots:=[x0,y0]:n:=100:
# Group 2, repeat n times. RK4 method.
for i from 1 to n do
 k1:=h*f(x0,y0):
 k2:=h*f(x0+h/2,y0+k1/2):
 k3:=h*f(x0+h/2,y0+k2/2):
 k4:=h*f(x0+h,y0+k3):
 Y:=y0+(k1+2*k2+2*k3+k4)/6:
 x0:=x0+h:y0:=Y:Dots:=Dots,[x0,y0];
od:
# Group 3, display some dots and plot.
 Dots[1],Dots[2],Dots[101];
 plot([Dots]);
```